

Chapitre 2 :

Le TABLEUR : FONCTIONS DE RECHERCHE,
CONTRÔLE ET ERGONOMIE DES FEUILLES

1 Chapitre 2 : Le TABLEUR : FONCTIONS DE RECHERCHE, CONTRÔLE ET ERGONOMIE DES FEUILLES

COMPETENCES ATTENDUES	SAVOIRS ASSOCIES
Automatiser des calculs en écrivant des formules	Découverte du tableur : Le classeur, la feuille de calcul, la cellule et les types de données
Concevoir et modifier la structure d'un modèle de feuille de calcul	Eléments d'ergonomie (formatage des cellules, gestion de l'affichage, validation des données, formatage conditionnel et gestion des erreurs)
Contrôler l'adéquation entre le contexte d'un problème de gestion et le modèle créé	
Mettre en place l'ergonomie d'une feuille de calcul	

1.1 Fonctions de recherche

Définition :

Les fonctions de recherche permettent de trouver la valeur d'une liste associée à une autre valeur placée dans une autre liste.

Exemple :

Une catégorie (codée A, B ou C) est affectée à chaque client. A chacune des catégories de la liste A- B – C correspond un taux de remise (dans l'ordre 0, 2 ou 4%). A partir du code de catégorie d'un client, la recherche dans la liste des codes permet de trouver le taux de remise à appliquer.

Les listes forment une table implantée au préalable dans une feuille de calcul active ou dans une autre feuille de calcul.

Les fonctions de recherche nécessitent, en général, que la première liste dans laquelle on cherche une valeur pour lui trouver une correspondance soit triée.

1.1.1 Fonction RECHERCHE (forme générale)

PRINCIPE :

La forme générale de la fonction RECHERCHE est :

=RECHERCHE(valeur à chercher ;colonne ou ligne de recherche ;colonne ou ligne de résultats)

La fonction RECHERCHE recherche une valeur dans une colonne (ou une ligne) d'un tableau et renvoie une valeur située à la même position dans la colonne (ou la ligne) qui contient les résultats.

Exemple :

Dans le tableau qui suit, le taux de remise du client « Optica » placé en C8 peut être déterminé directement, en fonction de la catégorie placée en C6, par la formule =RECHERCHE(C6 ;E3 :E5 ;F3 :F5)

	A	B	C	D	E	F
1	REMISES CLIENTS					
2	CODE CLIENT	NOM	CATEGORIE		CATEGORIE	REMISE
3	124	LE GALL	A		A	0%
4	132	HUMEAU	A		B	2%
5	329	SA LEGRAND	C		C	4%
6	543	OPTICA	B			
7						
8	REMISE OPTICA		2%			

Si la table CATEGORIES/REMISES était placée dans la feuille de calcul externe nommée CAT ci-après, la formule de recherche précédente deviendrait :

=RECHERCHE(C6 ;CAT !A2 :A4 ;CAT !B2 :B4)

	A	B
1	CATEGORIE	REMISE
2	A	0%
3	B	2%
4	C	4%

La feuille qui suit montre comment il est possible de calculer directement le montant de la remise en fonction du code du client placé en B10, à l'aide de **recherches imbriquées**.

B11								
=RECHERCHE(RECHERCHE(B10;A3:A6;C3:C6);E3:E5;F3:F5)*B9								
	A	B	C	D	E	F	G	H
1	REMISES CLIENTS							
2	CODE CLIENT	NOM	CATEGORIE		CATEGORIE	REMISE		
3	124	LE GALL	A		A	0%		
4	132	HUMEAU	A		B	2%		
5	329	SA LEGRAND	C		C	4%		
6	543	OPTICA	B					
7								
8	exemple calcul de remise							
9	VENTE	23166						
10	CLIENT	329						
11	REMISE	926,64						

En B11 on trouve : =RECHERCHE(RECHERCHE(B10;A3:A6;C3:C6);E3:E5;F3:F5)*B9

LISTES DISJOINTES

La fonction RECHERCHE permet si nécessaire, d'exploiter des données de listes ayant des directions différentes, horizontales ou verticales. Il est également possible d'effectuer des recherches par tranches puisque le tableur retient, par défaut, la catégorie correspondant au minimum de la tranche.

1.2 Fonctions RECHERCHEV et RECHERCHEH

La fonction RECHERCHEV joue le même rôle que la fonction RECHERCHE, mais elle effectue une recherche verticale, par colonnes. De même, la fonction RECHERCHEH recherche horizontalement par lignes.

LISIBILITE RENFORCEE

Les fonctions RECHERCHEV et RECHERCHEH, quand elles peuvent être mobilisées, permettent une expression plus rapide et plus lisible des formules, c'est pourquoi elles sont, le plus souvent préférées à la forme générale.

1.2.1 Fonction RECHERCHEV

La fonction RECHERCHEV permet de rechercher verticalement par l'index de la première colonne, à partir d'une valeur spécifiée, dans une table, la valeur correspondante dans une autre colonne indiquée et ce de façon exacte ou approximative.

Sa forme générale est la suivante :

=RECHERCHEV(valeur_recherchée ;table_matrice ;numéro_index_colonne,valeur_proche ;(vrai/faux)

Valeur que l'on va rechercher dans la première colonne de la table	Table de données dans laquelle on effectue la recherche	Numéro de la colonne dans laquelle se trouve la valeur que l'on recherche	Permet de faire une recherche exacte ou approchée
--	---	---	---

La table_matrice peut être nommée ou on peut y faire référence par une plage.

Le quatrième argument est « valeur_proche » (VRAI/FAUX) :

S'il est VRAI et que la valeur cherchée ne se trouve pas dans la liste ou que la première colonne n'est pas triée par ordre croissant (arrêt de la recherche dès qu'on atteint une valeur supérieure) alors c'est la première valeur immédiatement inférieure à la valeur recherchée qui sera retenue. Il permet éventuellement une recherche par tranche ce qui peut s'avérer utile en gestion.

S'il est FAUX, une recherche exacte sera effectuée. Si la valeur cherchée ne se trouve pas dans la liste ou que la première colonne n'est pas triée, la recherche ne pourra être effectuée et la fonction renverra le message d'erreur #N/A.

Exemple :

On souhaite retrouver le nom d'un salarié dans une table en fonction de son matricule.

B3 : `=RECHERCHEV(B2;D3:F5;2;FAUX)`

	A	B	C	D	E	F
1				table_matrice		
2	Matricule	HE-512		Matricule	Nom	Prénom
3	Nom	THOLAS		HE-511	HERBERT	L2A
4				HE-512	THOLAS	BERTRAND
5				FG-513	GARCIA	FLORE
6					Colonne 2 : données correspondantes	
7			Sens de la recherche			
8						

B3 : =RECHERCHEV(B2;\$D\$3:\$F\$5;2;FAUX)

Premier argument	Deuxième argument	Troisième argument	Quatrième argument
B2	\$D\$3:\$F\$5	2	FAUX
La valeur à chercher se trouve dans la cellule B2, soit ici HE-512	Table contenant les données	Index (rang) de la colonne où on trouve les valeurs de la donnée cherchée, ici 2 il s'agit donc des noms	La recherche se fera de façon exacte ce qui est obligatoire ici car les matricules ne sont pas classés par ordre alphabétique

Exemple :

Le taux de ristourne de l'Entreprise Gallo, client de l'entreprise Igloo, est obtenu comme suit :

B3 : `=RECHERCHEH(B2;E3:J4;2;VRAI)`

	A	B	C	D	E	F	G	H	I	J
1	Nom client	GALLO		Pourcentages de ristournes octroyées selon le montant total des chiffres d'affaires (intervalles de ces montants en colonnes ci-dessous)						
2	Chiffre d'affaires	125 000,00 €	Sens de la recherche							
3	Taux de ristourne	0,92%		Seuils	0	50 000	10 000	200 000	400 000	600 000
4	Montant de ristourne	1 150,00 €		Taux	0,75%	0,80%	0,92%	0,96%	0,97%	0,99%
5					Ligne 2 : données correspondantes					

Le quatrième argument doit être omis car il s'agit d'une recherche par tranches.

argument indique l'index de la ligne et non plus de la colonne à partir de laquelle la donnée correspondante est renvoyée.

CONSEIL

Lorsque vous utilisez la fonction RECHERCHEH la mobilisation des arguments est identique. Le troisième

METHODE

Principes fondamentaux des fonctions de recherche

- Veillez à ce que le premier argument recherché soit bien situé dans la première ligne ou première colonne de la

matrice indiquée au sein du deuxième argument.

- Dans certains contextes, il est nécessaire d'omettre le quatrième argument ou de mettre VRAI : travaux sur des dates, des intervalles (âge, tranches d'imposition ...)
- Dans d'autres contextes, il est indispensable d'utiliser le quatrième argument : Tri non croissant de la matrice du deuxième argument ou vérification d'une saisie effectuée par ailleurs.

1.3 Fonctions INDEX et EQUIV

Quand la recherche ne consiste ni à rechercher en fonction de l'index de la première colonne, ni en fonction de celui de la première ligne mais à l'intersection d'une ligne et d'une colonne, la fonction INDEX doit être mobilisée. Elle sélectionne une valeur dans un tableau à partir des index de numéros de ligne et de colonne, qui peuvent être paramétrés à l'aide des fonction EQUIV ou RECHERCHE.

1.3.1 Fonction INDEX

La forme générale de la fonction INDEX est la suivante :

=INDEX(Tableau ; no_ligne ; no_colonne)

Premier argument : Matrice (tableau) où l'on va rechercher une valeur	Deuxième argument : N° de la ligne	Troisième argument : N° de la colonne
Les argument 2 et 3 peuvent être retrouvés avec la fonction EQUIV		

Les arguments no_ligne et no_colonne doivent pointer sur une cellule appartenant à l'argument tableau. A défaut, la fonction INDEX renvoie la valeur d'erreur #REF !

Exemple :

Le prix de l'article Jupe chez Galopin peut être recherchée par une fonction INDEX :

B2	=INDEX(\$E\$3:\$F\$5;2;1)					
	A	B	C	D	E	F
1	Article	Jupe		Données		
2	Prix	45		Produit	Prix	Qté
3				Chemise	30	200
4			Ligne 2	Jupe	45	300
5				Pantalon	55	150
6					Colonne 1	

B2 : =INDEX(\$E\$3:\$F\$5;2;1)

Premier argument	Deuxième argument	Troisième argument
\$E\$3:\$F\$5	2	1

Table contenant les données	Numéro de la ligne de la table où se trouve les valeurs de la donnée cherchée, ici 2	Numéro de la colonne de la table où on trouve les valeurs de la donnée cherchée, ici 1
-----------------------------	--	--

1.3.2 Fonction EQUIV

La fonction EQUIV permet d'automatiser la détermination de la ligne ou de la colonne (deuxième et Troisième arguments) de la fonction INDEX.

EQUIV recherche une valeur spécifique dans une plage de cellules puis renvoie sa position relative (numéro de ligne ou de colonne) dans la plage.

La forme générale de la fonction EQUIV est la suivante :

=EQUIV(Valeur cherchée ; matrice recherche; type)

Valeur dont on recherche la position dans la liste	Liste dans laquelle se trouve la valeur	Argument facultatif dont le fonctionnement s'apparente à celui du quatrième argument de la fonction recherche
--	---	---

Le troisième argument, type est facultatif et peut prendre les valeurs suivantes :

- 0 Recherche une correspondance exacte
- 1 (ou omis) permet de chercher une valeur strictement inférieure ou égale à la valeur cherchée
- -1 permet de chercher une valeur strictement supérieure à la valeur cherchée
- 1 ou omis et -1 permettent des recherches par tranches.

Exemple :

La jupe de Galopin devient :

B2 X ✓ f_x =INDEX(\$E\$3:\$F\$5;EQUIV(B1;\$D\$3:\$D\$5;1);EQUIV(A2;\$E\$2:\$F\$2;1))									
	A	B	C	D	E	F	G	H	I
1	Article	Jupe		Données					
2	Prix	45		Produit	Prix	Qté			
3				Chemise	30	200			
4			Ligne 2	Jupe	45	300			
5				Pantalon	55	150			
6					Colonne 1				

Premier argument	Deuxième argument	Troisième argument
------------------	-------------------	--------------------

\$E\$3:\$F\$5	EQUIV(B1;\$D\$3:\$D\$5;1)	EQUIV(A2;\$E\$2:\$F\$2;1)
Table contenant les données	Automatise la détermination de la ligne	Automatise la détermination de la colonne
Les listes étant triées, le troisième argument des fonctions EQUIV peut être soit omis, soit égal à 1		

1.4 Imbrication des fonctions

Les fonctions RECHERCHEV Ou H, INDEX et EQUIV peuvent s'imbriquer et se compléter.

Une imbrication de RECHERCHEV est construite comme suit :

=RECHERCHEV(Valeur_cherchée ;Table_matrice ;Numéro_index_colonne ;valeur_proche(VRAI/FAUX))

Recherche Externe	Trouvée par une recherche interne	Arguments 2,3 et 4 de la recherche Externe
-------------------	--	--

RECHERCHEV(Valeur_cherchée ;Table ;Numéro_index_colonne ;VRAI/FAUX)

- Le résultat de la fonction RECHERCHEV la plus interne devient le premier argument de la fonction RECHERCHEV externe. Celle-ci détermine donc une valeur de ce premier argument
- La fonction RECHERCHEV externe effectue alors une seconde recherche dans une autre table à partir de la valeur trouvée par la recherche interne
- On peut également imbriquer plusieurs fonctions RECHERCHEV mais aussi des RECHERCHE, RECHERCHEH, EQUIV, INDEX ...

Exemple :

Détermination de la rémunération horaire en euros dans une feuille « intervention » :

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Personnel intervenant												
2	Numéro	Rémun E /h											
3	PERS1	27,00											
4	PERS2	62,00											
5	...												

La Recherche est effectuée à partir des tables PERS (feuille Personnel) et QUALIF (feuille Qualification) :

	A	B	C	D
1	Personnel intervenant			
2	Numéro	Nom	Prénom	Code Qualif
3	PERS1	JENSEN	Myriam	AC
4	PERS2	LOFOTEN	Caroline	DC
14	Zone A3:D13 nommée PERS			

	A	B	C
1	Qualification du personnel		
2	CODE QUALIF	LIBELLE	RENUMERATION HORAIRE
3	AC	Agent Commercial	27,00 €
4	DC	Directeur Commercial	62,00 €
13			
14	Zone A3:C12 nommée QUALIF		

Dans la cellule B3 de la feuille Intervention, on implante la formule suivante :

=RECHERCHEV(RECHERCHEV(A3;pers;EQUIV(Personnel!\$D\$2;Personnel!\$A\$2:\$D\$2;0));QUALIF;EQUIV(Qualification!\$C\$2;Qualification!\$A\$2:\$C\$2;0);FAUX)

La recherche interne

RECHERCHEV(RECHERCHEV(A3;pers;EQUIV(Personnel!\$D\$2;Personnel!\$A\$2:\$D\$2;0)) donne le code de la qualification (Code Qualif) du personnel dont le numéro est PERS1 (implanté en A3) soit AC.

La recherche externe détermine alors, à partir de ce code, la rémunération correspondante.

Les fonctions EQUIV permettent de retrouver les index de colonne pertinents pour chaque RECHERCHE.

1.5 Fonctions financières

Parmi les fonctions financières les plus courantes, retenons notamment :

- VA qui calcule la valeur actuelle
- TAUX qui calcule le taux d'intérêt
- NPM qui calcule le nombre de périodes
- VPM qui calcule le remboursement périodique d'un prêt.

Exemple :

Trois des quatre valeurs en B1, B2, B3 ou B4 doivent être connues. Nous précisons à chaque fois la formule à utiliser pour déterminer la quatrième valeur

	A	B	
1	Capital emprunté	20 000,00 €	$B1=VA(B2/12 ;B3 ;B4 ;0)$
2	Taux annuel	3,60%	$B2=TAUX(B3 ;B4 ;B1 ;0)*12$
3	Nombre de mois	48	$B3=NPM(B2/12 ;B4 ;B1)$
4	Mensualité	-448,01 €	$B4=VPM(B2/12 ;B3 ;B1 ;0 ;0)$

Les fonctions VAN (Valeur Actuelle Nette) et TRI (Taux de Rentabilité Interne) permettent d'évaluer les investissements à partir d'une succession de flux de trésorerie.

	A	B	C	D	E	F
1	Investissement	Flux 1	Flux 2	Flux 3		Taux
2	-10000	4000	4500	5000		8,00%
3	VAN	1531	$B3=VAN(F2;B2:D2)+A2$			
4	TRI	16%	$B4=TRI(A2:D2)$			

$VAN(TAUX ;VALEUR1 ; VALEUR2 ; ...)$

$TRI(VALEUR1 ; VALEUR2 ;...)$

1.6 Fonctions statistiques

Les fonctions dites statistique sont nombreuses :

Les principales

Fonctions	Exemple et résultat
ECARTYPE	=ECARTYPE(C5 :C28) Affiche l'écart type des valeurs de la plage C5 :C28
PENTE	=PENTE(A2 :F2 ;A1 :F1) Donne la pente (a) d'une droite d'ajustement d'équation $y = ax + b$ A2 :F2 → Valeurs y et A1 :F1 → Valeur x
TENDANCE	=TENDANCE(A2 :F2 ;A1 :F1 ;g1) Effectue une extrapolation linéaire et donne la valeur de y devant correspondre à la valeur de x contenue dans G1
COEFFICIENTCORRELATION	Donne le coefficient de corrélation de deux séries de valeurs
ORDONNEE.ORIGINE	Donne l'ordonnée d'origine (b) de la droite $y = ax + b$
DROITEG(utilisée de façon matricielle)	Donne la pente et l'ordonnée d'origine d'une droite d'ajustement

Figure 1 Principales fonctions statistiques

1.7 Ergonomie d'une feuille de calcul

1.7.1 Fonction d'information EST

Les fonctions EST sont utiles pour tester le contenu de cellules. Elles ne contiennent qu'un seul argument, obligatoire, qui est une valeur à tester.

L'argument valeur peut-être :

- Une valeur vide, c'est-à-dire une cellule qui ne contient aucune donnée ou formule et non qui n'affiche rien
- Une erreur renvoyée par le tableur
- Une valeur logique, de texte, de nombre ou une valeur de référence ou un nom s'y référant.

Quand la fonction EST n'est pas associée à une autre fonction elle renvoie VRA ou FAUX.

Les fonctions d'information les plus courantes sont au nombre de sept :

ESTVIDE	Renvoie vrai si l'argument valeur est vide
ESTERREUR	Renvoie vrai si l'argument valeur fait référence à une valeur d'erreur
ESTLOGIQUE	Renvoie vrai si l'argument valeur fait référence à une valeur logique
ESTNA	Renvoie vrai si l'argument valeur fait référence à une valeur d'erreur #N/A
ESTNUM	Renvoie vrai si l'argument valeur représente un nombre
ESTREF	Renvoie vrai si l'argument valeur est une référence
ESTTEXTE	Renvoie vrai si l'argument valeur se présente sous forme de texte

Figure 2 Utilité des principales fonctions EST

D'autres fonctions d'information sont accessibles et documentées dans l'aide du tableur.

Exemple :

L'amicale laïque utilise un tableur pour gérer ses salles de réunion.

	A	B	C	D
1	Catégorie Salle	Capacité d'Accueil	Capacité connue VRAI/FAUX	Capacité inconnue VRAI/FAUX
2	Secrétariat	Inconnu	FAUX	VRAI
3	Yoga	23	VRAI	FAUX
4	Arts plastiques	30	VRAI	FAUX
5	Comptabilité	inconnu	FAUX	VRAI

=ESTNUM(B2), implanté en C2, affiche VRAI si C2 contient une valeur numérique, FAUX, dans le cas contraire.

B2 contient du texte (inconnu) donc la fonction renvoie FAUX

A l'inverse, =ESTTEXTE(B2) renvoie VRAI si le contenu est du texte et donc inconnu et FAUX dans le cas contraire.

Les fonctions EST sont fréquemment associées à une alternative FONCTION SI pour renvoyer un message pertinent à l'utilisateur du tableur en cas d'erreur, d'absence de saisie et/ou pour effectuer un calcul sous condition.

Exemple :

Le modèle suivant a été implanté dans le tableur pour automatiser la création des factures de l'entreprise Rouillard. Ce modèle fait notamment référence à une table produit, les informations étant récupérées par des formules utilisant la fonction RECHERCHEV.

	A	B	C	D	E	F
1	FACTURE					
2	Numéro		Date		20%	
3	Référence	Désignation	quantité	HT	Total TTC	
4	SNO	Snow	1	147,85	177,42	
5	TIR	#N/A	12	#N/A	#N/A	
6		#N/A		#N/A	#N/A	

	A	B	C
1	Référence	Désignation	PUHT
2	ARY	Arys	7,20
3	BAR	Barathéon	27,41
4	CAS	Castral roc	201,25
5	CAT	Cateylen	85,36
6	SNO	Snow	147,85
7	STA	Stark	8,36
8	TYR	Tyrion	120,36
9	VAR	Varys	258,65
10	VEN	Vent gris	25,14
11	WIN	Winterfell	29,50
12			
13	Plage A2:C11 nommée PRODUITS		

De nombreuses erreurs apparaissent. Pour pallier le caractère peu professionnel du document, il est nécessaire de recourir aux fonctions d'informations :

	A	B	C	D	E	F	G	H	I	J	K	L
1	FACTURE											
2	Numéro		Date		20%							
3	Référence	Désignation	quantité	HT	Total TTC							
4	ARY	Arys	1	7,2	8,64							
5	TIR	Réf Erronée	12	Réf Erronée								

Les formules qui permettent de gérer les erreurs ou absences de saisies sont les suivantes :

```
B4=SI(ESTVIDE(A4);"";SI(ESTNA(RECHERCHEV(A4;PRODUITS;2;FAUX));"RéfErronée";RECHERCHEV(A4;PRODUITS;2;FAUX)))
```

```
D4=SI(ESTVIDE(A4);"";SI(ESTNA(RECHERCHEV(A4;PRODUITS;3;FAUX));"RéfErronée";RECHERCHEV(A4;PRODUITS;3;FAUX)))
```

```
E4=SI(ESTNUM(C4*D4);C4*D4;"")
```

1.7.2 Fonction SIERREUR

La fonction SIERREUR permet également de mettre en évidence les erreurs de calcul.

C'est une fonction logique, sa portée est plus générale et son usage plus simple que les fonctions d'information portant sur des erreurs car elle permet de traiter toutes les erreurs renvoyées par le tableur.

Sa syntaxe est la suivante :

```
=SIERREUR(valeur_testée ;valeur_si_erreur)
```

La valeur testée peut être une formule, une référence, une donnée ; elle est renvoyée s'il n'y a pas d'erreur. L'argument « valeur_si_erreur » peut prendre la forme d'un message affiché alertant l'utilisateur sur le type d'erreur.

Exemple :

La formule implantée en B3 dans le cas de l'entreprise Rouillard devient :

```
B4=SI(ESTVIDE(A4);"";SIERREUR(RECHERCHEV(A4;PRODUITS;2;FAUX);"Réf Erronée"))
```

1.7.3 Fonction CONCAT

La fonction CONCAT (anciennement CONTATENER) est une fonction de texte qui permet de regrouper plusieurs valeurs en une seule.

Définition :

Une **valeur** est un texte entre guillemets ou une formule ou une référence à une cellule contenant une formule.

Cette fonction permet de combiner, dans une seule chaîne de caractères, le résultat d'une formule et un texte explicatif. Elle est donc utile en gestion.

Sa syntaxe est la suivante :

```
=CONCAT(valeur1 ; valeur2 ; ...)
```

Exemple :

La formule implantée dans le cas Gallo en A6 permet de personnaliser par une phrase l’affichage du résultat des calculs pour le client dont le nom est en B1.

A6 =CONCAT("Le montant de la ristourne du client ";B1;" est de ";B4;" euros, pour un taux de ";RECHERCHEH(B2;E3:J4;2;VRAI))												
1	Nom client	GALLO		Pourcentages de ristournes octroyées selon le montant total des chiffres d'affaires (intervalles de ces montants en colonnes ci-dessous)								
2	Chiffre d'affaires	125 000,00 €	Sens de la recherche									
3	Taux de ristourne	0,92%		Seuils	0	50 000	10 000	200 000	400 000	600 000		
4	Montant de ristourne	1 150,00 €		Taux	0,75%	0,80%	0,92%	0,96%	0,97%	0,99%		
5				Ligne 2 : données correspondantes								
6	Le montant de la ristourne du client GALLO est de 1150 euros, pour un taux de 0,0092											

A6 =CONCAT("Le montant de la ristourne du client ";B1;" est de ";B4;" euros, pour un taux de ";RECHERCHEH(B2;E3:J4;2;VRAI))

L’opération de **concaténation** peut contenir au maximum 255 éléments et un total de 8192 caractères.

1.7.4 Formats de cellules


Excel offre de nombreuses possibilités pour adapter le format des cellules aux documents de gestion à produire en fonction du contexte.

Le menu ACCUEIL regroupe l’essentiel des possibilités offertes :

- Polices de caractères (type, style, taille, couleur, soulignement...)
- Alignement (Orientation, positionnement, renvoi à la ligne, fusionner les cellules ...)
- Nombre (Standard, monétaire, comptabilité, date personnalisée ...)
- Bordure (Affichage, remplissage, personnalisé ...)
- Style (Affichage conditionnel, tableaux prédéfinis, styles prédéfinis ...)
- Cellules (Hauteur ligne, largeur colonne, couleur des onglets)
- Protection et tris

Exemple :

Voici les règles de formatage conditionnel des notes obtenues au DCG.



	A	B	C	D	E	L	M	N
1	DCG RESULTATS AUX EXAMENS							
2	ANNEE	2016						
3	Coefficient	1	1	1	1	1,5	1	1
4		UE1	UE2	UE3		UE11	UE12	UE13
5	ETU 1	8	12	11,75		15	12,5	11,75
6	ETU 2	8	8,75	10,25		14,5	5	11,25
7	.../...							
8	ETU 40	10	10	10		14,5	11,25	8,25

1.7.5 Validation des données

Il est possible d'ajouter des **règles de validation** des données pour une cellule ou une plage de cellules. Elles sont regroupées dans le menu « DONNEES /Validation des données ».

On peut spécifier le type de données saisies (décimales, liste de valeurs autorisées, dates, pour un intervalle de validation...).

Un message personnalisé peut être associé à la saisie.

Des messages d'alerte peuvent également être sélectionnés.

1.8 SYNTHÈSE

Le tableur : Fonctions de recherche, contrôle et ergonomie des feuilles

Les fonctions de recherche

Forme générale	=RECHERCHE(valeur à chercher ;liste où l'on recherche ;liste de résultats)
Recherche verticale par colonnes	=RECHERCHEV(Valeur_cherchée ;Table_matrice ;numéro_index_colonne ;valeur_proche (VRAI/FAUX))
Recherche horizontale par lignes	=RECHERCHEH(Valeur_cherchée ;Table_matrice ;numéro_index_ligne ;valeur_proche (VRAI/FAUX))
Recherche à l'intersection d'une ligne et d'une colonne	=INDEX(Tableau ;no_ligne ;no_colonne)

Les fonctions de recherche spécifiques

Les numéros des colonnes ou des lignes peuvent être déterminés par la fonction EQUIV :

=EQUIV(valeur_cherchée ;matrice_recherche ;type)

Les fonctions RECHERCHEV, RECHERCHEH, INDEX et EQUIV peuvent s'imbriquer et se compléter.

Les fonctions de test logique et la validation des données :

- Les fonctions EST sont utiles pour tester le contenu de cellules et permettre la gestion des erreurs, combinées à la fonction SI et à un message d'erreur pertinent. Elles améliorent l'ergonomie des feuilles de calcul
- La fonction logique SIERREUR joue le même rôle avec une portée plus générale
- Les formats doivent être adaptés au contexte et répondre à des exigences de qualité du document qui doit jouer un rôle professionnel.
- La validation des données permet également d'améliorer l'ergonomie et de limiter les erreurs de saisie